
AUTOSTART_EDAS.LST
AC1 U880 Assembler - Source Listing
 
              0010 ;*******************************************
              0020 ;*Autostart - Speichern eines Programms auf*
              0030 ;*Kassette (mit optionalem Startbild) und  *
              0040 ;*selbststaendigem Start nach dem Einlesen *
              0050 ;*-----------------------------------------*
              0060 ;*Programm        von B.Nickel,   (c) 1988 *
              0070 ;*Rueckuebersetzt von J.Spannenkrebs, 2017 *
              0080 ;*******************************************
              0090 ;
              0100 ; Beschreibung des Verfahrens
              0110 ;
              0120 ; Daten in Bloecken zu 256 Byte + Pruefsumme
              0130 ; Geschwindigkeit 1500 bit/s
              0140 ;
              0150 ; Bitformat:
              0160 ; 1 Bit --> 666 mikrosek lang
              0170 ;    _         _
              0180 ;  _|  --> 1,   |_ --> 0
              0190 ; Byteformat:
              0200 ; MSB zuerst,  <-- |7|6|5|4|3|2|1|0|7|. usw.
              0210 ;
              0220 ; Format der Aufzeichnung:
              0230 ;
              0240 ;*******************************************
              0250 ; Vorspann:
              0260 ; 512 Nullbytes zum Einschwingen der
              0270 ; Aussteuerungsautomatik
              0280 ;
              0290 ; 0E6H, Synchronisationsbyte
              0300 ;
              0310 ; 055H, Kennbyte fuer Namen
              0320 ;
              0330 ; Namen, max. 16 Zeichen,
              0340 ; Rest wird mit 20H aufgefuellt
              0350 ;
              0360 ; 256 Nullbytes, Zeit fuer Namensdarstellung
              0370 ; beim Einlesen
              0380 ;
              0390 ;*******************************************
              0400 ; optional kann ein Startbild vor den
              0410 ; eigentlichen Daten eingefuegt werden
              0420 ;
              0430 ; 03BH, als Kennung fuer Programm mit Bild
              0440 ;
              0450 ; dann 8 Datenbloecke mit Bildinhalt:
              0460 ; 03CH, Kennung Datenblock
              0470 ; 0xxH, Blocklaenge, 256 Byte
              0480 ; 0llH, Blockstartadresse, Lowbyte
              0490 ; 0hhH, Blockstartadresse, Highbyte
              0500 ; ...., 256 Datenbyte
              0510 ; 0xxH, Pruefsumme:
              0520 ; 8-Bit-Summe von High- und Lowteil der
              0530 ; Blockstartadresse sowie der 256 Datenbyte,
              0540 ; Ueberlauf wird nicht beruecksichtigt
              0550 ;
              0560 ; weitere 7 Datenbloecke
              0570 ;
              0580 ;*******************************************
              0590 ; nun folgen Programm bzw. Daten,
              0600 ;
              0610 ; 03AH, als Kennung Programm ohne Bild,
              0620 ; (bei Programm mit Bild wird diese Kennung
              0630 ; nicht ausgegeben)
              0640 ;
              0650 ; dann weiter mit
              0660 ; Datenbloecken mit Programminhalt:
              0670 ; 03CH, Kennung Datenblock
              0680 ; 0xxH, Blocklaenge, normal 256
              0690 ; Restblock entsprechend weniger
              0700 ; 0llH, Blockstartadresse, Lowbyte
              0710 ; 0hhH, Blockstartadresse, Highbyte
              0720 ; ...., bis zu 256 Datenbyte

              0730 ; 0xxH, Pruefsumme:
              0740 ; 8-Bit-Summe von High- und Lowteil der
              0750 ; Blockstartadresse sowie der bis zu 256
              0760 ; Datenbyte,
              0770 ; Ueberlauf wird nicht beruecksichtigt
              0780 ;
              0790 ; eventuell weitere Datenbloecke
              0800 ;
              0810 ;*******************************************
              0820 ; 03FH, Kennung Autostart
              0830 ;
              0840 ; Datenblock mit der Autostartroutine:
              0850 ; 03CH, Kennung Datenblock
              0860 ; 011H, Blocklaenge = 17
              0870 ; 0C0H, Blockstartadresse, Lowbyte
              0880 ; 018H, Blockstartadresse, Highbyte
              0890 ; ...., 17 Datenbyte
              0900 ; 0xxH, Pruefsumme:
              0910 ; 8-Bit-Summe von High- und Lowteil der
              0920 ; Blockstartadresse sowie der 16 Datenbyte,
              0930 ; Ueberlauf wird nicht beruecksichtigt
              0940 ;
              0950 ; Datenblock
              0960 ; mit Startadresse der Autostartroutine:
              0970 ; 03CH, Kennung Datenblock
              0980 ; 002H, Blocklaenge = 2
              0990 ; 006H, Blockstartadresse, Lowbyte
              1000 ; 018H, Blockstartadresse, Highbyte
              1010 ; ...., 2 Datenbyte
              1020 ; 0xxH, Pruefsumme:
              1030 ; 8-Bit-Summe von High- und Lowteil der
              1040 ; Blockstartadresse sowie der 2 Datenbyte,
              1050 ; Ueberlauf wird nicht beruecksichtigt
              1060 ;
              1070 ;*******************************************
              1080 ; 078H, Kennung Ende
              1090 ;
              1100 ; 0llH, Startadresse, lowbyte
              1110 ; 0hhH, Startadresse, highbyte
              1120 ;
              1130 ;*******************************************
              1140 ; Die Kennungen 03AH, 03BH und 03FH kennt
              1150 ; die Leseroutine im Monitor nicht.
              1160 ; Die Kennungen sind fuer Kopier- und
              1170 ; Listprogramme noetig um die Adressen
              1180 ; richtig auszuwerten.
              1190 ;*******************************************
              1200 ; Programm ist selbstmodifizierend
              1210 ; laeuft also nur im RAM
              1220 ;*******************************************
              1230 ;
              1240 ; Vereinbarungen
              1250 ; --------------
0000          1260        ORG  0
              1270 ;
0000 0083     1280 CTRLC  EQU  03H+80H   ;CTRL-C gedrueckt
0000 0020     1290 SPACE  EQU  20H       ;Leerzeichen
              1300 ;
              1310 ; Kennbyte fuer:
0000 00E6     1320 SYNCHR EQU  0E6H      ;Synchronisation
0000 0055     1330 NAMKEN EQU  055H      ;Namen
0000 003A     1340 PRGKEN EQU  03AH      ;nur Programm
0000 003B     1350 BLDKEN EQU  03BH      ;Bild + Programm
0000 003C     1360 BLKKEN EQU  03CH      ;(Daten)Block
0000 003F     1370 AUTKEN EQU  03FH      ;Autoststart
0000 0078     1380 ENDKEN EQU  078H      ;Ende
              1390 ;
              1400 ; Zeitkonstanten (bei 2 MHz-Takt)
0000 0031     1410 ZK1    EQU  49        ;Laenge 1.Halbwelle
0000 002E     1420 ZK2    EQU  46        ;Laenge 2.Halbwelle
              1430 ;
0000 07FD     1440 GETCO1 EQU  007FDH    ;Monitorschleife
              1450 ;
0000 1800     1460 CURPOS EQU  01800H    ;Cursorposition

              1470 ;
0000 1806     1480 RRST10 EQU  01806H    ;RST10H-Umleitung
              1490 ; in RAM
              1500 ;
0000 18C0     1510 FRERAM EQU  018C0H    ;freier System-RAM
              1520 ;
0000 1000     1530 BWSANF EQU  01000H    ;Bildschirmanfang
0000 17FF     1540 BWSEND EQU  017FFH    ;Bildschirmende,
0000 10BF     1550 LIUN3Z EQU  010BFH    ;links, unten,
              1560 ; 3.Zeile
              1570 ;
0000 E000     1580 PICRAM EQU  0E000H    ;Anfang Bild im RAM
              1590 ;
              1600 ; Argumente:
0000 185B     1610 ARG1   EQU  0185BH    ;Anfangsadresse
0000 185D     1620 ARG2   EQU  0185DH    ;Endadresse
0000 185F     1630 ARG3   EQU  0185FH    ;Startadresse
              1640 ;
              1650 ; PIO-Adressen:
0000 0004     1660 PIOADA EQU  4         ;PIO Kanal-A, Tastatur
              1670 ;
0000 0005     1680 PIOBDA EQU  5         ;PIO Kanal-B, Kassette
              1690 ;
              1700 ;*******************************************
E800          1710        ORG  0E800H
              1720 ;
              1730 ; Prolog Autostart ohne Bild
              1740 ; a ANFANG ENDE START NAMEN
              1750 ;
E800 0009610D 1760        DEFB 0,9,"a,0DH
              1770 ;
              1780 ; Modus Programm ohne Bild
E804 3E11     1790        LD   A,11H
E806 1806     1800        JR   AUTO01
              1810 ;
              1820 ; Prolog Autostart mit Bild
              1830 ; Bild liegt von 0E000H bis 0E7FFH im RAM
              1840 ; b ANFANG ENDE START NAMEN
              1850 ;
E808 0009620D 1860        DEFB 0,9,"b,0DH
              1870 ;
              1880 ; Modus Programm mit Bild
E80C 3E12     1890        LD   A,12H
              1900 ;
              1910 ; Modus merken
E80E 3227E9   1920 AUTO01 LD   (MODUS),A
              1930 ;
              1940 ; DE auf Anfang Namen stellen
E811 1B       1950 AUTO02 DEC  DE
E812 1A       1960        LD   A,(DE)
E813 FE20     1970        CP   SPACE     ;auf Anfang Namen
E815 28FA     1980        JR   Z,AUTO02
              1990 ;
E817 2A5F18   2000        LD   HL,(ARG3) ;Startadresse
E81A 2220E9   2010        LD   (PRGSTA),HL
              2020 ; in Autostartroutine eintragen
              2030 ;
              2040 ; Vorton, 512 Nullbytes
E81D 0600     2050        LD   B,0       ;256
E81F AF       2060 AUTO03 XOR  A         ;Nullbyte
E820 CD9FE8   2070        CALL BYTOUT    ;ausgeben
E823 AF       2080        XOR  A         ;256 Nullbyte
E824 CD9FE8   2090        CALL BYTOUT    ;ausgeben
E827 10F6     2100        DJNZ AUTO03    ;256 x 2 = 512
              2110 ;
              2120 ; Synchronisation
E829 3EE6     2130        LD   A,SYNCHR  ;Synchronbyte
E82B CD9FE8   2140        CALL BYTOUT    ;ausgeben
E82E 3E55     2150        LD   A,NAMKEN  ;Kennbyte Namen
E830 CD9FE8   2160        CALL BYTOUT    ;ausgeben
              2170 ;
              2180 ; Namen ausgeben
E833 0610     2190        LD   B,16      ;16 stelligen Namen
E835 1A       2200 NAMOUT LD   A,(DE)    ;vom BWS holen

E836 CD9FE8   2210        CALL BYTOUT    ;und ausgeben
E839 1B       2220        DEC  DE
E83A 10F9     2230        DJNZ NAMOUT
              2240 ;
              2250 ; kurze Pause 256 Nullbytes
E83C 0600     2260        LD   B,0       ;256
E83E AF       2270 AUTO04 XOR  A         ;Nullbyte
E83F CD9FE8   2280        CALL BYTOUT    ;ausgeben
E842 10FA     2290        DJNZ AUTO04
              2300 ;
              2310 ; MODUS - mit oder ohne Bild auswerten
E844 3A27E9   2320        LD   A,(MODUS)
E847 FE11     2330        CP   11H       ;ohne Bild ?
E849 2814     2340        JR   Z,AUTO05  ;ja, dann Programm
              2350 ;                      nein, erst Bild
              2360 ;
E84B 3E3B     2370        LD   A,BLDKEN  ;Kennung Programm
              2380 ; mit Bild
E84D CD9FE8   2390        CALL BYTOUT    ;ausgeben
              2400 ;
              2410 ; Bild ausgeben
              2420 ; Bild liegt ab 0E000H im RAM
E850 210010   2430        LD   HL,BWSANF ;von Anfang BWS
E853 11FF17   2440        LD   DE,BWSEND ;bis Ende BWS
E856 FD2100E0 2450        LD   IY,PICRAM ;Anfang Bild im RAM
E85A CDC8E8   2460        CALL BLKOUT    ;ausgeben
E85D 1805     2470        JR   AUTO06
              2480 ;
E85F 3E3A     2490 AUTO05 LD   A,PRGKEN  ;Kennung Programm
              2500 ; ohne Bild
E861 CD9FE8   2510        CALL BYTOUT    ;ausgeben
              2520 ;
              2530 ; Programm ausgeben
              2540 ; Argumente Anfang und Ende auswerten
E864 2A5B18   2550 AUTO06 LD   HL,(ARG1) ;Anfang Programm
E867 ED5B5D18 2560        LD   DE,(ARG2) ;Ende Programm
E86B FD2A5B18 2570        LD   IY,(ARG1) ;Quelle Anfang
              2580 ; Programm im RAM
E86F CDC8E8   2590        CALL BLKOUT    ;Programm ausgeben
              2600 ;
              2610 ; Autostartroutine ausgeben
E872 3E3F     2620        LD   A,AUTKEN  ;Kennung Autostart
E874 CD9FE8   2630        CALL BYTOUT
              2640 ;
              2650 ; Ziel Anfang, freier Speicher ab 18C0H
E877 21C018   2660        LD   HL,FRERAM
E87A 11D018   2670        LD   DE,FRERAM+16 ;18D0H, Ziel Ende
E87D FD2114E9 2680        LD   IY,MANIPU ;Quelle Anfang
E881 CDC8E8   2690        CALL BLKOUT
              2700 ;
              2710 ; Zeichenausgabe manipulieren
E884 210618   2720        LD   HL,RRST10 ;RST10H patchen
E887 110718   2730        LD   DE,RRST10+1
E88A FD2125E9 2740        LD   IY,AUTOAD ;Autostartadresse
E88E CDC8E8   2750        CALL BLKOUT
              2760 ;
              2770 ; Endekennung und Startadresse ausgeben
E891 3E78     2780        LD   A,ENDKEN  ;Endekennzeichen
E893 CD9FE8   2790        CALL BYTOUT
E896 2A5F18   2800        LD   HL,(ARG3) ;Startadresse
E899 CD02E9   2810        CALL WRDOUT    ;Wort ausgeben
              2820 ;
              2830 ; zurueck zur Monitorschleife
E89C C3FD07   2840        JP   GETCO1
              2850 ;*******************************************
              2860 ;
              2870 ; ein Byte ausgeben
              2880 ; -----------------
              2890 ; Ausgabe als Stream auf PIO Kanal-B, Bit6
              2900 ; hoechstwertigstes Bit zuerst
              2910 ; Byteformat:
              2920 ; MSB zuerst, <-- |7|6|5|4|3|2|1|0|7|.. usw.
              2930 ; Bitformat:
              2940 ; 1 Bit --> 666 mikrosek lang,

              2950 ;   _         _
              2960 ; _|  --> 1,   |_ --> 0
              2970 ; Beispiel 0E6H
              2980 ; Bit     | 7 | 6 | 5 | 4 | 3 | 2 | 1 | 0 |
              2990 ; Daten   | 1 | 1 | 1 | 0 | 0 | 1 | 1 | 0 |
              3000 ;            _   _   _ _   _     _   _ _
              3010 ; Stream  x_| |_| |_|   |_| |_ _| |_|   |_x
              3020 ; erste Halbwelle ist zum Datenbit negiert
              3030 ;
E89F C5       3040 BYTOUT PUSH BC
E8A0 4F       3050        LD   C,A       ;Datenbyte nach C
E8A1 37       3060        SCF
              3070 ; C enthaelt die Daten und ist gleichzeitig
              3080 ; Bitzaehler,
              3090 ;
E8A2 CB11     3100 BITLOP RL   C         ;Carry nach Bit 0
              3110 ; hoechstwertigstes (Daten)Bit nach Carry
              3120 ; und Test ob das Byte 0 ist
E8A4 2819     3130        JR   Z,BYTEND  ;wenn 0 dann fertig
              3140 ; sonst Bit ausgeben, dazu
E8A6 DB05     3150        IN   A,(PIOBDA) ;PIO-Daten lesen
E8A8 CBF7     3160        SET  6,A       ;Bit 6 auf high
E8AA 3002     3170        JR   NC,BITLOW ;war Datenbit 1?
E8AC CBB7     3180 BITHIG RES  6,A       ;nein, Bit 6 low
E8AE D305     3190 BITLOW OUT  (PIOBDA),A ;Ausgabe 1.Halbw.
              3200 ;
E8B0 0631     3210        LD   B,ZK1     ;Zeitkontante 1
E8B2 10FE     3220 ZK1LOP DJNZ ZK1LOP    ;herunterzaehlen
              3230 ;
E8B4 C640     3240        ADD  A,40H     ;Flankenwechsel
E8B6 D305     3250        OUT  (PIOBDA),A ;Ausgabe 2.Halbw.
              3260 ;
E8B8 062E     3270        LD   B,ZK2     ;Zeitkonstante 2
E8BA 10FE     3280 ZK2LOP DJNZ ZK2LOP    ;herunterzaehlen
E8BC A7       3290        AND  A         ;Carry auf 0
              3300 ; fuer Bitzaehler, dann weitere
E8BD 18E3     3310        JR   BITLOP    ;Datenbit ausgeben
              3320 ;
E8BF C1       3330 BYTEND POP  BC
E8C0 DB04     3340        IN   A,(PIOADA) ;Tastatur einlesen
E8C2 FE83     3350        CP   CTRLC     ;CTRL-C gedrueckt?
E8C4 CAFD07   3360        JP   Z,GETCO1  ;ja, Abbruch
E8C7 C9       3370        RET
              3380 ;
              3390 ; Daten in Bloecke zerlegen und ausgeben
              3400 ; --------------------------------------
              3410 ; Aufruf:
              3420 ; DE zeigt auf Ende Quelldaten
              3430 ; HL zeigt auf Anfang Quelldaten
              3440 ; IY zeigt auf Anfang Quelldaten im RAM
              3450 ;
              3460 ; waehrend der Abarbeitung
              3470 ; in B wird die Blocklaenge berechnet
              3480 ; in C wird die Pruefsumme berechnet
              3490 ; HL zeigt auf Zieladr. des akt. Datenbyte
              3500 ; IY zeigt auf Quelladr. des akt. Datenbyte
              3510 ;
E8C8 D5       3520 BLKOUT PUSH DE        ;DE, Ende auf Stack
              3530 ;
              3540 ; Anzahl der Bloecke berechnen
E8C9 EB       3550        EX   DE,HL     ;HL=Ende, DE=Anfang
E8CA A7       3560        AND  A         ;Carry = 0
E8CB ED52     3570        SBC  HL,DE     ;HL = Ende - Anfang
              3580 ;                      Ueberlauf?
              3590 ; in H stehen die Anzahl der Bloecke
              3600 ; in L stehen die Anzahl der Restbytes
              3610 ;
E8CD 3831     3620        JR   C,BLKEND  ;ja, dann beenden
              3630 ;
              3640 ; Blockkennung ausgeben
E8CF 3E3C     3650        LD   A,BLKKEN  ;Blockkennung
              3660 ;
              3670 ; Blocklaenge korrigieren,
              3680 ; Ende - Anfang ergibt immer 1 Byte zuwenig

E8D1 23       3690        INC  HL
E8D2 CD9FE8   3700        CALL BYTOUT    ;ausgeben
              3710 ;
              3720 ; Blocklaenge berechnen
E8D5 7C       3730        LD   A,H       ;Blockanzahl
E8D6 A7       3740        AND  A         ;gleich 0?
E8D7 2804     3750        JR   Z,BLKVAR  ;ja, dann Restlaenge
              3760 ;
              3770 ; nein, dann ganzer Block
E8D9 0600     3780        LD   B,0       ;B = 256 BYTE
E8DB 1801     3790        JR   BLK256    ;ganzer Block
              3800 ;
E8DD 45       3810 BLKVAR LD   B,L       ;Restlaenge nach B
              3820 ;
E8DE EB       3830 BLK256 EX   DE,HL     ;HL=Anfang, DE=Ende
E8DF D1       3840        POP  DE        ;Ende vom Stack holen
              3850 ;
              3860 ; Blocklaenge ausgeben
E8E0 78       3870        LD   A,B       ;Blocklaenge
E8E1 CD9FE8   3880        CALL BYTOUT    ;ausgeben
              3890 ;
E8E4 7D       3900        LD   A,L       ;Blockstart-Lowadr.
E8E5 84       3910        ADD  A,H       ;+ Blockst.-Highadr.
E8E6 4F       3920        LD   C,A       ;= Anfangswert in C
E8E7 CD02E9   3930        CALL WRDOUT    ;Blockstartadr. ausg.
              3940 ;
              3950 ; einen Block ausgeben
              3960 ; Laenge in B
              3970 ; IY zeigt auf zu uebertragende Daten
              3980 ;
              3990 ; Pruefsumme ueber Daten berechnen
E8EA 79       4000 BLKLOP LD   A,C       ;Pruefsumme nach A
E8EB FD8600   4010        ADD  A,(IY)    ;Datenbyte addieren
E8EE 4F       4020        LD   C,A       ;neue Pruefs. nach C
              4030 ;
              4040 ; Datenbyte ausgeben
E8EF FD7E00   4050        LD   A,(IY)    ;Datenbyte nach A
E8F2 CD9FE8   4060        CALL BYTOUT    ;und ausgeben
              4070 ; Adresszaehler erhoehen
E8F5 23       4080        INC  HL        ;Anfangsadresse + 1
E8F6 FD23     4090        INC  IY        ;Datenadresse + 1
              4100 ;
E8F8 10F0     4110        DJNZ BLKLOP    ;bis Block fertig
              4120 ;
              4130 ; Pruefsumme ausgeben
E8FA 79       4140        LD   A,C       ;Pruefsumme nach A
E8FB CD9FE8   4150        CALL BYTOUT    ;und ausgeben
              4160 ;
E8FE 18C8     4170        JR   BLKOUT    ;weitere Bloecke
              4180 ;
E900 D1       4190 BLKEND POP  DE        ;alle Bloecke durch
E901 C9       4200        RET
              4210 ;
              4220 ; Wort ausgeben
              4230 ; -------------
E902 7D       4240 WRDOUT LD   A,L
E903 CD9FE8   4250        CALL BYTOUT    ;Ausgabe Low-Byte
E906 7C       4260        LD   A,H
E907 CD9FE8   4270        CALL BYTOUT    ;Ausgabe High-Byte
E90A C9       4280        RET
              4290 ;
              4300 ; Hilfsroutine
              4310 ; ------------
              4320 ; naechste 3 Zeilen werden nach Laden des
              4330 ; Autostartmakers ausgefuehrt wenn er mit
              4340 ; mit sich selbst abgespeichert wurde,
              4350 ; ohne Bild -> a E800 E927 E90B NAME oder
              4360 ; mit Bild  -> b E800 E927 E90B NAME,
              4370 ;
              4380 ; Damit wird das Startbild aber nur auf
              4390 ; dem Bildschirm sichtbar.
              4400 ; Um das Startbild auch in RAM ab 0E000H zu
              4410 ; laden sind folgende Aufrufe noetig:
              4420 ; ohne Bild -> a E000 E927 E90B NAME oder

              4430 ; mit Bild  -> b E000 E927 E90B NAME,
              4440 ;
              4450 ; das Bild wird dabei zweimal abgespeichert,
              4460 ; einmal mit Zieladresse BWS und einmal
              4470 ; mit Zieladresse 0E000H vor dem Programm
              4480 ;
              4490 ; Anfang 3. Zeile von unten
E90B 21BF10   4500 AUTOST LD   HL,LIUN3Z
              4510 ;
              4520 ; Cursor definiert positionieren
              4530 ; damit Bild nicht hochgescrollt wird
E90E 220018   4540        LD   (CURPOS),HL
              4550 ;
              4560 ; zurueck in Monitorschleife
E911 C3FD07   4570        JP   GETCO1
              4580 ;
              4590 ; eigentliche Autostartroutine
              4600 ; ----------------------------
              4610 ; wird in RAM ab 18C0H geladen
              4620 ; wird bei naechster Zeichenausgabe ueber
              4630 ; RST10H ausgefuehrt
              4640 ;
              4650 ; RST10H wieder als Zeichenausgabe
E914 211300   4660 MANIPU LD   HL,13H
E917 220618   4670        LD   (RRST10),HL
              4680 ;
              4690 ; "Lade-Sternchen" durch SPACE ersetzen
E91A 2A0018   4700        LD   HL,(CURPOS)
E91D 3620     4710        LD   (HL),SPACE
              4720 ;
              4730 ; geladenes Programm starten
E91F CD       4740        DEFB 0CDH      ;CALL
E920 0BE9     4750 PRGSTA DEFW AUTOST    ;Autostartadresse
              4760 ;
              4770 ; wenn Programm beendet wird dann
              4780 ; zurueck in die Monitorschleife
E922 C3FD07   4790        JP   GETCO1
              4800 ;
E925 C018     4810 AUTOAD DEFW FRERAM    ;Startadresse der
              4820 ; Autostartroutine
              4830 ;
E927 0001     4840 MODUS  DEFS 1         ;Merker
              4850 ;
 
 
AC1 U880 Assembler - Symbol Table
 
185BH 1610 ARG1         185DH 1620 ARG2
185FH 1630 ARG3         003FH 1370 AUTKEN
E80EH 1920 AUTO01       E811H 1950 AUTO02
E81FH 2060 AUTO03       E83EH 2270 AUTO04
E85FH 2490 AUTO05       E864H 2550 AUTO06
E925H 4810 AUTOAD       E90BH 4500 AUTOST
E8ACH 3180 BITHIG       E8A2H 3100 BITLOP
E8AEH 3190 BITLOW       003BH 1350 BLDKEN
E8DEH 3830 BLK256       E900H 4190 BLKEND
003CH 1360 BLKKEN       E8EAH 4000 BLKLOP
E8C8H 3520 BLKOUT       E8DDH 3810 BLKVAR
1000H 1530 BWSANF       17FFH 1540 BWSEND
E8BFH 3330 BYTEND       E89FH 3040 BYTOUT
0083H 1280 CTRLC        1800H 1460 CURPOS
0078H 1380 ENDKEN       18C0H 1510 FRERAM
07FDH 1440 GETCO1       10BFH 1550 LIUN3Z
E914H 4660 MANIPU       E927H 4840 MODUS
0055H 1330 NAMKEN       E835H 2200 NAMOUT
E000H 1580 PICRAM       0004H 1660 PIOADA
0005H 1680 PIOBDA       003AH 1340 PRGKEN
E920H 4750 PRGSTA       1806H 1480 RRST10
0020H 1290 SPACE        00E6H 1320 SYNCHR
E902H 4240 WRDOUT       0031H 1410 ZK1
E8B2H 3220 ZK1LOP       002EH 1420 ZK2
E8BAH 3280 ZK2LOP
 
        ***** 0000 Errors *****

